

SINTAXE DA LINGUAGEM MINI-TRIANGLE

(Páginas 8 e 9 de PLPJ)

```
Program ::=
    single-Command
single-Command ::=
    V-name := Expression
    | Identifier ( Expression )
    | if Expression then single-Command
      else single-Command
    | while Expression do single-Command
    | let Declaration in single-Command
    | begin Command end
Command ::=
    single-Command
    | Command ; single-Command
Expression ::=
    primary-Expression
    | Expression Operator primary-Expression
primary-Expression ::=
    Integer-Literal
    | V-name
    | Operator primary-Expression
    | ( Expression )
V-name ::=
    Identifier
Identifier ::=
    Letter
    | Identifier Letter
    | Identifier Digit
Letter ::=
    a | b | c | ... | z
Digit ::=
    0 | 1 | 2 | ... | 9
Integer-Literal ::=
    Digit
    | Integer-Literal Digit
Operator ::=
    + | - | * | / | < | > | =
Declaration ::=
    single-Declaration
    | Declaration ; single-Declaration
single-Declaration ::=
    const Identifier ~ Expression
    | var Identifier : Type-denoter
Type-denoter ::=
    Identifier
Comment ::=
    ! CommentLine EOL
CommentLine ::=
    Graphic CommentLine
EOL ::=
    end-of-line character
Graphic ::=
    any printable character or space
```